' Encoding - © 1993-2003 Aiman Hanna

' This program explores strings and some of the operations that can be performed over them. 

' The program also illustrates how arrays and strings can be related.

' The program performs string encoding based on a predetermined encoding method. The performed

' encoding is case insensitive; that is both ‘A’ and ‘a’ will have the same encoded character.

' Key Points:

'       1) Strings and arrays      

'       2) Ucase

Option Explicit

Option Base 1

Const MaxLen As Integer = 37 ' 26 characters, space and 10 digits

Const RegularStr As String = "ABCDEFGHIJKLMNOPQRSTUVWXYZ0123456789 "

Const EncodingStr As String = "CZXVBMN123QWRETyIUOP654JLKHFG DAS0978"

Dim CodeArr(1 To 2, 1 To MaxLen) As String ' Array of two strings

Private Sub Form_Load()

    Dim i As Integer

    ' Put the regular character string in the first row of the array

    For i = LBound(CodeArr, 2) To UBound(CodeArr, 2)

        CodeArr(1, i) = Mid(RegularStr, i, 1)

    Next i

    ' Put the encoding character string in the second row of the array

    For i = LBound(CodeArr, 2) To UBound(CodeArr, 2)

        CodeArr(2, i) = Mid$(EncodingStr, i, 1)

    Next i

End Sub

Private Sub cmdPerformCoding_Click()

    Dim i As Integer

    Dim InpStr As String, OutStr As String, InpCh As String, OutCh As String

    Dim Pos As Integer

    OutStr = ""

    ' Read the input string

    InpStr = txtInput.Text

    ' For each character in the input string find the encoding character

    For i = 1 To Len(InpStr)

        InpCh = Mid(InpStr, i, 1)

        InpCh = UCase(InpCh)    ' convert it to upper case

        ' find the position of that character in the regular string

        Pos = InStr(RegularStr, InpCh)

        ' The encoded character is the character at the same position in the array

        OutCh = CodeArr(2, Pos)

        OutStr = OutStr & OutCh

    Next i

    txtOutput.Text = OutStr

End Sub

Private Sub cmdExit_Click()
    End

End Sub

